
Introduction

Lorsqu'il faut traiter des valeurs dans des variables similaires qui prennent la
forme X1-Xn, le premier réflexe consiste souvent a associer ces variables a
un vecteur et a utiliser une boucle DO pour arriver a nos fins.

Peux-on penser utiliser d'autres stratégies, et méme des stratégies peut-étre
plus efficientes pour obtenir les mémes résultats?

Nous pensons que oui et que nous pouvons y arriver sans l'utilisation de
vecteurs et de boucles mais plutét avec des fonctions et aussi par une
approche inhabituelle qui fait appel a un fichier factice avec ou sans format.

@] & oy

Voyons ce qu'il en est.



Il était une fois...

Nous voulons rechercher une lésion spécifique suite a un accident (de travail
ou de la route) parmi un maximum de 4 lésions pour 5 individus. Un code
de lésion se définit sur 3 caracteres.

Pour y arriver nous allons utiliser respectivement la méthode usuelle consistant
a utiliser un vecteur et une boucle DO et une méthode faisant appel a des
fonctions notamment celles de |la famille "CAT".

Ensuite nos voulons trouver les individus qui sont affectés par I'une ou l'autre
des lésions définies dans un contexte ou les codes de lésions peuvent avoir
de 3 a 5 caracteres. Les 3 premiers caractéres identifient la Iésion principale
et les 2 autres si présents spécifient un niveau de précision plus grand pour
ces lésions.

Pour ce dernier exemple, nous allons procéder progressivement en faisant
appel tour a tour aux fonctions, a un fichier factice et a une combinaison
format+fichier.



Recherche codes de lésions
(cas simples)

Table SAS source: Lesions
Variables: identifiant et 4 codes de lésions
Valeurs: codes de 3 caracteres

025 027 240
101 210 220 240
024 260 250
025 104 260 800
250 923 240



Recherche codes de lésions
Approche 1 avec vecteur+boucle

Trouver les lésions 240 ou 250 et créer
un indicateur booléen 0/1 pour chacun

data Lesions_Appl(drop=i);
set Lesions;
array Lx(4);
c250=0; c240=0;
do 1=1 to dim(Lx);

1T Lx(n) =: "250" then c250 = 1;
1T Lx(n) =: "240" then c240 = 1;
end;

run,




Recherche codes de léesions
Résultat 1 avec vecteur+boucle

Table créée: Lesions App1

025 027 |2
101 210 220 |240 0
024 260 (250 1
0
1

025 104 260 800
2500 923 | 240

Résultat: correct avec cette approche



Recherche codes de léesions
Approche 2 avec fonctions

Utiliser la fonction CAT (4

data Lesions_App2;
set Leslions;
Concat=cat(of Lx:);
c250=fFind(Concat,"250") > O;
c240=find(Concat,'240'") > O
run;

Variable intermédiaire Concat non nécessaire ici mais
utile pour illustrer le résultat de la concaténation



Recherche codes de léesions
Résultat 2 avec fonctions

Table créée: Lesions App2

Obs. id  Lx1 Lx2 Lx3 Lx4 Concat  c250 c240
25 0p7 240 (125027240

101 210 220 240 101210220240 0
024 260 250 024260250
025 104 260 800 025104260800
250 923 240 250923240

1
1
0
0
1

e Py

Résultat: incorrect avec la fonction CAT



Recherche codes de lésions
Approche 3 avec fonctions

Utiliser la fonction CATX avec un astérisque comme séparateur

data Lesions_ App3;
set Lesions;
Concat=catx("*",of Lx:);
c250=find(Concat, ""*250") > O;
c240=find(Concat, ""*240") > O;
run;




Recherche codes de léesions
Résultat 3 avec fonctions

Table créée: Lesions App3

025 027 240 025*027*240
101 210 220 240 101*210*220*240 0
024 260 250 024*260*250 1
025 104 260 800 025*104*260*800 0
250] 923 240 250923*240 (0

—5 =) | e | | e

Résultat incorrect avec la fonction CATX et un séeparateur ™"



Recherche codes de lésions
Approche 4 avec fonctions

Utiliser la fonction CATX avec un espace comme séparateur

data Lesions_App4;
set Lesions;
Concat=catx(" ",of Lx:);
c250=find(Concat, "'250"") > O;
c240=find(Concat, ''240") > O;
run;




Recherche codes de léesions
Résultat 4 avec fonctions

Table créée: Lesions App4

025 027 240 025 027 240

101 210 220 240 101 210 220 240 0

024 260 250 024 260 250 1

025 104 260 800 025 104 260 800 0

250] 923 240 250923 240 @

.
1
0
0
1

Résultat correct avec la fonction CATX, et le séparateur " "



Recherche codes de lésions
Approche 5 avec fonctions

Utiliser la fonction CATX avec un astérisque comme séparateur
et I'ajout en d'un "|" comme élément a concaténer

data Lesions_App5;
set Lesions;
Concat=catx("*", "|",of Lx:);
c250=find(Concat, ""*250") > O;
c240=find(Concat, ""*240') > O;
run;

Cette approche non nécessaire ici va s'avérer utile plus loin



Recherche codes de léesions
Résultat 5 avec fonctions

Table créée: Lesions App5

025 027 240 *025*027*240
101 210 220 240 |*101*210*220*240
024 260 250 *024*260*250
025 104 260 800 |*025*104*260*800
250( 923 240 *25071923*240

@ -0

Wk

Résultat correct avec la fonction CATX, un et le séparateur



Recherche codes de lesions +
(cas plus complexes)

Table SAS source: LesionsPlus
Variables: identifiant et 4 codes de lésions+
Valeurs: codes de 3 a 5 caracteres
Code lésion: 3 caract. et sous-lésion: +1 ou 2 caract.

0251 027 [24009
101 21044 220 24009
02435 26002 [25099
02507 104 26002 800
250| 92388 [2407

On cherche les personnes avec le code 240xx ou 250xx



Recherche codes de lésions +
Approche 1 avec fonctions

Utiliser la fonction CATX avec un espace comme séparateur

data LesionsPlus Appl;
set LesionsPlus;
Concat=catx(" ", of Lp:);

c250 = find(Concat,'250") > 0O;
c240 = find(Concat,''240") > O;
run;

Pour I'approche "vecteur+boucle" on peut utiliser le code
SAS vu au tout début pour créer la table Lesions_App1



Recherche codes de lesions +

Résultat 1 avec fonctions

Table créée: LesionsPlus App1

0251 027 24005 0251 027 24005
101 21044 220 24009 10121044 220 24009
02435 26002 25099 02435 26002 25099
02507] 104 26002 800 104 26002 800
250 92388 2407 250 92388 2407

Résultat: incorrect avec la fonction CATX et le séparateur

0
1

9

1

1
1
0
0
1



Recherche codes de lésions +
Approche 2 avec fonctions

Utiliser la fonction CATX avec un astérisque comme séparateur

data LesionsPlus App2;
set LesionsPlus;
Concat=catx(""*", of Lp:);
c250 = find(Concat,"*250") > 0O;
c240 find(Concat, ""*240") > 0O;
run;

Comme dans les approches avec fonctions
pour des données un peu plus complexes



Recherche codes de lesions +
Résultat 2 avec fonctions

Table créée: LesionsPlus App2

0251 027 24005 0251*027*24005 1
101 21044 220 24009 101*21044*220*24009 0 1
02435 26002 25099 02435*26002*25099 1 0
02507 104 26002 800  02507*104*26002*800 0 0
200 92388 2407 250192388*2407 @ 1

TEA 1]

Résultat: incorrect avec la fonction CATX et le séparateur



Recherche codes de lésions +
Approche 3 avec fonctions

Utiliser la fonction CATX avec un astérisque comme séparateur
et I'ajout en d'un "|" comme élément a concaténer

data LesionsPlus App3;
set LesionsPlus;
Concat=catx(""*", "|'"", of Lp:);
c250 = find(Concat,""*250") > O;
c240 = find(Concat,'*240") > 0O;
run;




Recherche codes de lesions +
Résultat 3 avec fonctions

Table créée: LesionsPlus App3

D ABC 0251 027 24005 ¥0251*027124005 1
IV DEF 101 21044 220 24009 [*101*21044*220124009 0 1
=l GHI 02435 26002 25099 [*02435*26002{25099 1 0
26002 800  |*02507*104*26002*800 0 0

250 92388 2407 *25092388*2407 1 1

Résultat: correct avec la fonction CATX, un "|" et le séparateur "*"



Recherche codes de lesions +
Approche 4 avec fichier et fonctions

Utilisation détournée d'un fichier factice
Remarquer le format $3. utilisé ici \

filename bidon dummy; ‘
data LesionsPlus App4,;

set LesionsPlus;
length buffer $ 16;
file bidon;

put (Lp:) ($3. +1) @;

buffer=_file_ ;

c250 = find( _file ,""250") > O;

c240 find(_file ,"240") > 0
put;

run;

L 'utilisation d'un fichier entraine la création de la
variable _fi1le_ qu'on exploite ici



Recherche codes de lesions +
Résultat 4 avec fichier et fonctions

Table créée: LesionsPlus App4

0251 027 24005 025 027[240 0 1
101 21044 220 24009 101210 220[240 0 1
02435 26002 25099 024 260 250 1 0
0 0
1 1

02507 104 26002 800 025 104 260 800
250 92388 2407 1250{923|240

Résultat: correct avec I'énoncé PUT et un format $3.
Le format $3. ne considere que les 3 premiers caractéres




Recherche codes de lésions +
Approche 5 avec format et fichier

Etape 1 de 2:
Création d'un format pour les intervales requis

proc format;
value $LpFmt

"240" - "'24099" = X"
"250" - '"25099" = "Y"
other = "@";

run,

On couvre les valeurs entre 2x0 a 2x099 pour créer des
indicateurs X, Y ou @



Recherche codes de lésions +
Approche 5 avec format et fichier

Etape 2 de 2:
Utilisation d'un fichier factice et de I'énoncé PUT pour
Trouver individus avec les lésions transformées en X et Y

filename bidon dummy;

data LesionsPlus App5;
length Pattern $4; /* Sinon 32k */
set LesionsPlus;
file bidon;
put (Lp:) (SLpFmt.) @;
Pattern = Ffile_;

*1T findc( _file ,"XY'") then output;
put;

run;

Le IF peut étre activé pour n'obtenir que les observations désiréees



Recherche codes de lesions +
Résultat 5 avec format et fichier

Table créée: LesionsPlus_App5

0251
101

250

027
21044

02435 26002
02507 104

92388

24005

220

24009

25099

26002

2407

800

@eX@
@e@X
@QeY@
Qree@
Y@X@

Pattern indique de gauche a droite la présence ou non d'une des
valeurs recherchées dans une position correspondant aux variables
Lp1 a Lp4. Le "@" signifie qu'on a pas trouvé une de ces valeurs.



Synthese

Tableau synthese de quelques approches
possibles de recherche de valeurs
avec ce qui fonctionne "Oui" et ce qui ne fonctionne pas "Non"

plus

Vecteur+Boucle Intuitif et traditionnel

CAT Non Non Fonctionne nulle part
CATX et ™" Non Non Fonctionne nulle part
CATX et"" Oui Non Piege a éviter

CATX et "" et "|" Oui Oui Intéressant

File et format $3. Oui Oui Inusité et innovateur

File et format SLpFmt. Oui Oui Innovateur et souple

XXX : Couvert dans la présentation



Conclusion

Les stratégies de recherche présentées sont valables et acceptables dans la
plupart des situations. Celles impliquant fichier avec ou sans format sont
moins évidentes a premiere vue.

L'approche conventionnelle avec vecteur est souvent celle a laquelle nous
avons le réflexe de recourir. Elle nécessite sans doute un peu plus de temps
de traitement en raison de I'exécution itérative de code engendrée par la
boucle DO.

Une autre approche propose de voir les choses autrement et d'utiliser la
puissance et la versatilité des fonctions pour arriver a nos fins. Elle est un
peu plus concise et directe que la premiére mais demande une attention
particuliere pour obtenir de bons résultats. Quelques piéges sont a éviter.

Les 2 derniéres approches proposent un détour par l'utilisation du "buffer"
d'écriture _file_ afin de pouvoir utiliser I'énoncé PUT avec ou sans un format
crée.

Il vous reste a choisir I'approche qui vous convient parmi celles démontrées ou
par d'autres approches que vous pourriez inventer! Pourquoi pas?

Questions ?
N Commentaires ?
Suggestions ?

A

L'astigmatisme entraine une vision floue en raison de la déformation de la cornée qui
ne fait pas bien converger les rayons lumineux vers le méme point dans la rétine.

Ainsi certaines fagons d'utiliser SAS peuvent apparaitre floues au départ mais avec les
lunettes de I'apprentissage et de I'expérimentation tout devient plus clair.

Contact:

1 data _null ; d (:) b

2 courriel=lowcase(cat(reverse("Alain"),reverse("VOyeR"), "@gmail.com"));
3 putlog "NOTE: " courrlel—
4 run:

NOTE: courriel=niala-@gmail.com

Marques de commerce

SAS et tous les autres noms de produits et services de SAS Institute Inc. sont des marques de commerce de SAS Institute Inc. aux Etats-Unis et
dans les autres pays. ® indique I'inscription au Etats-Unis. Les autres noms de produits ou marques sont des marques déposées ou des marques
de commerce de leurs compagnies respectives.

*------------------------------------------------------;

* Club SAS de Québec, 3 juin 2021                      ;

*------------------------------------------------------;



* Présentation

  Titre: "Comment rechercher des valeurs de données en SAS®"

  Sous-titre: "Approches diverses entre tradition et innovation"

  Par Alain Voyer, Analyste en informatique et formateur SAS pigiste

;



options nodate center nonumber;

title;

footnote;



* Partie 1 - Exemples de recherches de valeurs;



* Scénario: Codes simples de lésions subies

 (accident de travail ou accident de la route ou ...)

;



* Recherche de valeurs dans une table SAS;



* Données sources;

* Liste des codes de lésions d`un individu;

* Codes simples sur 3 positions;



data Lesions;

  infile datalines truncover;

  input id $3. (Lx1-Lx4) (: $3.);

datalines;

ABC 025 027 240

DEF 101 210 220 240

GHI 024 260 250

JKL 025 104 260 800

MNO 250 923 240

;

run;



title1 "Données sources: Liste des codes de lésions d'un individu";

title1 color=black "Table SAS source: " color=blue "Lesions";

title2 color=black "Variables: " color=blue "identifiant et 4 codes de lésions";

title3 color=black "Valeurs: " color=blue "codes de 3 caractères";



proc print;

run;



* Présence des lésions 240 ou 250 avec indicateur;

* Approche 1 avec vecteur et boucle;



data Lesions_App1(drop=i);

  set Lesions;

  array Lx(4);

  c250=0; c240=0;

  do i=1 to dim(Lx);

    if Lx(i) =: "250" then c250 = 1;

    if Lx(i) =: "240" then c240 = 1;

  end;

run;



title1 color=black "Table créée: " color=green "Lesions_App1";

proc print;

run;



* Présence des lésions 240 ou 250 - Approche 2 avec fonctions;



data Lesions_App2;

  set Lesions;

  Concat=cat(of Lx:);

  c250=find(Concat,"250") > 0;

  c240=find(Concat,"240") > 0;

run;



title1 color=black "Table créée: " color=red "Lesions_App2";

proc print;

run;



* Présence des lésions 240 ou 250 - Approche 2 avec fonctions;



data Lesions_App3;

  set Lesions;

  Concat=catx("*",of Lx:);

  c250=find(Concat,"*250") > 0;

  c240=find(Concat,"*240") > 0;

run; 



title1 color=black "Table créée: " color=red "Lesions_App3";

proc print;

run;



* Présence des lésions 240 ou 250 - Approche 3 avec fonctions;



data Lesions_App4;

  set Lesions;

  Concat=catx(" ",of Lx:);

  c250=find(Concat,"250") > 0;

  c240=find(Concat,"240") > 0;

run; 



title1 color=black "Table créée: " color=green "Lesions_App4";

proc print;

run;



* Présence des lésions 240 ou 250 - Approche 5 avec fonctions;



data Lesions_App5;

  set Lesions;

  Concat=catx("*","|",of Lx:);

  c250=find(Concat,"*250") > 0;

  c240=find(Concat,"*240") > 0;

run;



title1 color=black "Table créée: " color=green "Lesions_App5";

proc print;

run;


*------------------------------------------------------;

* Club SAS de Québec, 3 juin 2021                      ;

*------------------------------------------------------;



* Présentation

  Titre: "Comment rechercher des valeurs de données en SAS®"

  Sous-titre: "Approches diverses entre tradition et innovation"

  Par Alain Voyer, Analyste en informatique et formateur SAS pigiste

;



options nodate center nonumber;

title;

footnote;



* Partie 2 - Exemples de recherches inusitées;



* Scénario: Codes plus complexes de lésions subies

 (accident de travail ou accident de la route ou ...)

;



* Données sources;

* La lésion comporte 3 caractères et, s`il y

  a lieu, suvis de 1 ou 2 autres caractères qui

  caractérisent plus précisément la lésion

;



data LesionsPlus;

  infile datalines truncover;

  input id $3. (Lp1-Lp4) (: $5.);

datalines;

ABC 0251 027 24005

DEF 101 21044 220 24009

GHI 02435 26002 25099

JKL 02507 104 26002 800

MNO 250 92388 2407

;

run;



title1 color=black "Table SAS source: " color=blue "LesionsPlus";

title2 color=black "Variables: " color=blue "identifiant et 4 codes de lésions+";

title3 color=black "Valeurs: " color=blue "codes de 3 à 5 caractères";

title4 color=black "Code lésion: " color=blue "3 caract. et sous-lésion: +1 ou 2 caract.";



proc print;

run;



* Compare 2 stratégies: traditionnelle et avec fichier;

* Approche 1 avec fonction CATX;



data LesionsPlus_App1;

  set LesionsPlus;

  Concat=catx(" ", of Lp:);

  c250 = find(Concat,"250") > 0;

  c240 = find(Concat,"240") > 0;

run;



title1 color=black "Table créée: " color=red "LesionsPlus_App1";

proc print; run;



* Approche 2 avec fonction CATX;



title1 "LesionsPlus_App2";

data LesionsPlus_CATX_Err2;

  set LesionsPlus;

  Concat=catx("*", of Lp:);

  c250 = find(Concat,"*250") > 0;

  c240 = find(Concat,"*240") > 0;

run;



title1 color=black "Table créée: " color=red "LesionsPlus_App2";

proc print;

run;



* Approche 3 avec CATX;



title1 "LesionsPlus_App3";

data LesionsPlus_CATX_Ok;

  set LesionsPlus;

  Concat=catx("*","|", of Lp:);

  c250 = find(Concat,"*250") > 0;

  c240 = find(Concat,"*240") > 0;

run;



title1 color=black "Table créée: " color=green "LesionsPlus_App3";

proc print;

run;



* Approche 4 avec fichier factice et fonctions;



filename bidon dummy;



title1 "LesionsPlus_App4";

data LesionsPlus_File_Ok;

  set LesionsPlus;

  length buffer $ 16;

  file bidon;

  put (Lp:) ($3. +1) @;

  buffer=_file_;

  c250 = find(_file_,"250") > 0;

  c240 = find(_file_,"240") > 0;

  put;

run;



title1 color=black "Table créée: " color=green "LesionsPlus_App4";

proc print;

run;



* Approche 5 Alternative avec format et fichier;



proc format;

  value $LpFmt

    "240" - "24099" = "X" 

    "250" - "25099" = "Y"

    other           = "@";

run;



* Approche avec format, fichier bidon et fonction;



filename bidon dummy; /* Longueur du buffer _file_ = 32 767 */



data LesionsPlus_App5;

  length Pattern $4; /* Sinon 32k */

  set LesionsPlus;

  file bidon;

  put (Lp:) ($LpFmt.) @;

  Pattern = _file_;

 *if findc(_file_,"XY") then output;

  put;

run;



title1 color=black "Table créée: " color=green "LesionsPlus_App5";

proc print; 

  var id Lp: Pattern;

run;



* Approche 6 alternative avec fonctions et format;

* Résultat équivalent au précédent (approche 5);

*   (mais sans l`observation no. 4);

* Nécessite de répéter les PUT cependant;



data LesionsPlus_App6;

  set LesionsPlus;

  Pattern=catt(put(Lp1,$LpFmt.)

              ,put(Lp2,$LpFmt.)

              ,put(Lp3,$LpFmt.)

              ,put(Lp4,$LpFmt.)

              );

  if findc(Pattern,"XY") then output;

run;



title1 color=black "Table créée: " color=green "LesionsPlus_App6";

proc print; 

  var id Lp: Pattern;

run;





